

MAT 1010

Innføring i bruk av Maple

Vårsemesteret 2005

Dette er en liten øvelse i bruk av Maple. Den kan gjennomføres på egenhånd på en av fakultetets datamaskiner.

Lykke til!

DEL 1

1. Sett i gang Maple på din arbeidsstasjon. Etter en kort stund kommer det frem et Maple-arbeidsdokument på din skjerm. Inntil videre heter det "Untitled(1)".

Det er meningen at du skal skrive de aller fleste kommandoene som blir omtalt videre inn i dette dokumentet. Du vil også bli bedt om å løse noen enkle oppgaver.

Øverst i venstre hjørne ser du tegnet [$>$] etterfulgt av en markør som blinker. Dette betyr at Maple venter på at en kommando skal tastes in.

Når du skriver inn en kommando, **avslutter den med semi-kolon tegnet ;** og trykker **Return**, vil kommandoen bli utført og Maple's svar kommer rett under. Under svaret vil du da se et nytt tegn [$>$] etterfulgt av den blinkende markøren. Dette betyr at Maple venter på en ny kommando.

La oss utføre en enkel utregning. De fire vanlige operasjonene med tall, addisjon, subtraksjon, multiplikasjon og divisjon kodes ved symbolene $+$, $-$, $*$ og $/$ i Maple. Det er fort gjort å glemme å skrive $*$ når man skal gange sammen to tall / uttrykk, men Maple vil stort sett gi beskjed om at noe er galt.

Vi regner ut $\frac{2 \cdot 7 + 4 \cdot 13 - \frac{19 \cdot 5}{7}}{6}$ ved

```
[ > ( 2*7 + 4*13 - 19*5/7 ) / 6;
```

Maple bruker samme prioriteringsrekkefølge mellom de forskjellige operasjonene som du antagelig er vant fra bruk av kalkulator. Men er du i tvil er det alltid lurt å spandere noen paranteser.

2. Dersom vi **avslutter kommandoen med kolon-tegnet :** i stedet for

semikolon-tegnet ; og trykker Return blir kommandoen effektivert av Maple, men svaret vises ikke.

Dette er ofte nyttig når man skal be Maple laste inn noen tilleggspakker som den ikke laster inn fra start (for å ikke bruke opp maskinhukommelsen unødvendig). Det er også nyttig for å utføre en mellomregning der svaret blir brukt i neste kommando. Vi kan nemlig alltid **henvisse til forrige svar** ved å skrive inn **prosent-tegnet %** i neste kommando. Vi kan da gjerne skrive inn begge kommandoene i samme linje. La oss se på et eksempel:

```
[ > ( 2*7 + 4*13 - 19*5/7 ) / 6:  
[ > % - 17/42;  
[ >
```

Alternativt kan vi skrive

```
[ > ( 2*7 + 4*13 - 19*5/7 ) / 6: % - 17/42;  
[ >
```

Vi kan nå bruke dette svaret videre:

```
[ > % * 108/5;  
[ >
```

3. Dersom det er noe galt i kommandoen som man ber Maple om å effektivere, kommer Maple med en **feilmelding**, og markøren blinker der en mulig feil kan være.

Det første du bør sjekke er om feilen ikke bare skyldes en ren skrivefeil. Det vil det ofte være i begynnelsen. Er det ikke det kan du prøve å utnytte den hjelpen Maple tilbyr ved å dra ned **Hjelp**-rullgardinen.

En veldig vanlig feil er som sagt å glemme gangetegnet * når man ganger sammen to tall/uttrykk. La oss prøve å regne 99 ganget med forrige svar ved å "glemme" * tegnet:

```
[ > 99%;
```

Markøren blinker nå på din skjerm mellom 99 og % fordi Maple tror at feilen er der. Maple kan forøvrig ikke vite at vi ønsket å gange sammen her, altså at det vi ønsket å skrive er 99*%.

Legg merke til at når markøren blinker der en mulig feil kan være, kommer det **ikke** et nytt [> under feilmeldingen: det er meningen at du skal rette opp feilen og trykke Return for å effektivere den korrekte kommandoen uten å måtte skrive inn alt på nytt.

Retter du nå opp feilen i kommandoen ovenfor på din skjerm får du frem

```
[ > 99 * %;  
[ >
```

Dersom du ikke klarer å rette opp en feil og ønsker å gå videre med en annen kommando, drar du ned "Insert"-rullgardinen, aktiviserer "Execution group" og velger enten "Before Cursor" eller "After Cursor" (avhengig av hvor markøren din

står plassert).

Alternativt kan du viske vekk hele kommandoen og feilmeldingen (på vanlig måte).

Etterhvert bør du på egenhånd prøve ut de mulighetene som rullgardinene og knapperaden tilbyr. De er mange, men vi skal bare omtale det som er strengt nødvendig. Med en liten innsats vil du oppdage at Maple kan også brukes som et tekstbehandlingsprogram der matematiske formler og uttrykk kan inngå. Denne øvelsen er feks. skrevet ved å bruke Maple som skriveverktøy.

Oppgave 1 Betrakt (den ukorrekte) kommandoen $[> 9 (1 - 2/3 + 4*5 - 8/7;$

a) Effektuer denne. Tenk over hva som er feilen(e).

b) Lag et nytt $[>$ under feilmeldingen.

c) Skriv inn den korrekte kommandoen og effektuer denne. (Det kan her bruke kopieringsmulighetene hvis du vil spare krefter!)

d) Gå tilbake til den ukorrekte kommandoen og visk vekk både den og den tilhørende feilmeldingen. Du kan gjerne bruke "**Delete paragraph**" kommandoen som finnes på **Edit**-rullgardinen.

4. I alle kommandoene vi har skrevet hittil har vi bare brukt rasjonale tall og Maple oppgir da svaret som et rasjonalt tall. **Dersom det inngår minst et tall på desimalform vil Maple oppgi svaret på desimalform.** Som regel oppgir Maple svaret med opptil 10 desimaler. Dette kan man forandre på hvis man vil, men vi kommer til å klare oss lenge med 10 desimaler.

La oss som et eksempel regne ut 1.2734^{10} og tredjeroten av 100.

Vi trenger da å vite at uttrykket x^a **kodes som x^a i Maple.** Merk at vi må bruke parenteser rundt a dersom a ikke består av et enkelt tall/bokstav

Det første går greit:

```
[ > 1.2734^10;
```

Det neste går ikke fullt så greit:

```
[ > 100^(1/3);
```

```
[ >
```

Maple synes tydeligvis at dette er svaret bra nok! Sannheten er imidlertid den at Maple prøver å beregne tredje roten av 100 som et rasjonalt tall (siden vi oppga bare rasjonale tall i kommandoen. Den klarer ikke det og derfor returneres bare uttrykket.

Vi kan imidlertid **tvinge Maple til å beregne et talluttrykk på desimalform** ved å gi kommandoen

[> **evalf(uttrykket);**

Et alternativ er å oppgi et av tallene på desimalform.

Her kan vi velge enten

```
[ > evalf(%);  
[ >
```

eller

```
[ > evalf(100^(1/3));  
[ >
```

eller

```
[ > 100.0^(1/3);  
[ >
```

Vi kan sjekke at svaret er riktig ved å utføre

```
[ > %^3;  
[ >
```

Oppgave 2 a) Beregn $2.72^{(-4)}$ b) Beregn $3^{\left(\frac{1}{10}\right)}$.

5. Mange av de vanlige konstantene og funksjonene finnes ferdig lagret i Maple.

Vi ser på noen eksempler:

a) Vi prøver å beregne $\tan\left(\frac{\pi}{6}\right) + \cos\left(\frac{\pi}{9}\right)$ ved

```
[ > tan(Pi/6) + cos(Pi/9);  
[ >
```

Maple vil gjerne oppgi svarene på en "pen" form, men får det ikke til når det gjelder å beregne $\cos\left(\frac{\pi}{9}\right)$.

Vi får da svaret på desimalform ved å effektuere

```
[ > evalf(%);  
[ >
```

b) Vi ønsker å beregne $\ln(3 + 5e^{(-2)})$, men er usikre om Maple klarer å finne en pen form for svaret. Vi kan da like godt skrive følgende to kommandoer på samme linje (svarene kommer rett under hverandre):

```
[ > ln( 3+ 5*exp(-2) );      evalf( ln( 3+ 5*exp(-2) ) );
[ >
```

Legg merke til at vi skrev inn $\exp(-2)$ for $e^{(-2)}$ og ikke $e^(-2)$. Spesielt bør det huskes at tallet e skrives som $\exp(1)$.

Oppgave 3:

a) Beregn $\frac{5 \sin\left(\frac{\pi}{12}\right)}{6} - 3 \cos(1.2 \pi)$.

b) Beregn $\ln\left(\left|\frac{\pi}{2} - 3\right|\right) - \frac{e}{\pi^2}$.

6. Maple kan også brukes til å regne med **symbolske uttrykk**, der det inngår en eller flere variable, slik som polynomer og trigonometriske funksjoner. Variablene i slike uttrykk kalles ofte x , y eller z av gammel vane, men du kan bruke (nærmest) hva du vil av bokstavkombinasjoner.

For å forsøke å **utvikle et produkt** kan man bruke kommando

```
[> expand( (produkt-)uttrykk );
```

Vi utvikler f.eks produktuttrykket $(x + 2y + 3z)(x - y + 4z)$ ved

```
[ > expand( (x+2*y+3*z)*(x-y+4*z) );
```

Vi kan få Maple til å beregne et uttrykk ved å **angi verdien for en eller flere av variablene** som inngår. Vi bruker da **subs**-kommando

```
[> subs( x=verdi1, y= verdi2, ..., uttrykk i x,y,...);
```

For å finne verdien av ovenstående uttrykk når $x = 2$ og $y = 3$ (som et uttrykk i variabelen z) kan vi derfor effektuer

```
[ > subs( x=2, y=3, % );
```

Verdien av dette uttrykket når $z = 2.1$ får vi frem ved

```
[ > subs( z=2.1, % );
```

Maple "kan" binomialteoremet, dvs utvikle produktet $(a + b)^n$. Dette kan vi sjekke når

n=10 ved å effektuere

```
[ > expand( (a+b)^10 );  
[ >
```

Vi kan videre beregne $(2 + b)^{10}$ ved

```
[ > subs( a=2, % );
```

eller mere direkte ved

```
[ > expand( (2+b)^10 );
```

Vi tar med noen eksempler til som viser at "expand"-kommandoen "virker" bra i noen andre tilfeller (her de to første) men ikke gir noe nytt i noen andre (her det siste).

```
[ > expand( exp(x+y) );  
[ > expand( sin(x-y) );  
[ > expand( ln(x*y) );
```

7. Noen ganger ønsker vi å gjøre det motsatte av å utvikle et produkt: vi lurer på om et uttrykk kan skrives som et produkt, mao om uttrykket kan faktoriseres. Vi kan la Maple prøve å gjøre dette ved å gi kommandoen

```
[> factor(uttrykk);
```

Ofte vil ikke Maple klare å faktorisere det du ber om, dette er mye vanskeligere enn å ekspandere et uttrykk.

Vi prøver å faktorisere $a^4 - b^4$ ved

```
[ > factor(a^4-b^4);  
[ >
```

Legg merke til at Maple i utgangspunktet antar at variablene er *reelle* variable. Den faktorerer derfor ikke uttrykket $a^2 + b^2$ i svaret ovenfor.

Maple tilbyr også kommandoene

```
[> combine(uttrykk);      og      [> simplify(uttrykk);
```

Disse kan brukes hhvis til å kombinere og forenkle enkelte uttrykk, som f.eks. trigonometriske uttrykk. Vær obs på at Maple's svar ikke alltid vil være slik som du

forventer, for det er ikke alltid lett for den å avgjøre hva som forventes av en forenkling.

Vi tar med noen eksempler :

```
[ > combine(sin(x)*sin(y));  
[ > combine(sin(3*x)*cos(4*y));  
[ > simplify(cos(3*x)-cos(x));  
[ > combine(%);  
[ > simplify(3-4*sin(x)^2+ 10*cos(x)^2);  
[ > simplify(3*cos(2*x)-4*cos(x)^2);
```

Oppgave 4:

a) Ekspander $(x + 2y)(3y - 4z)(x - z)^2$.

Angi uttrykket som en funksjon x når $y = -1$ og $z = 4$.

b) Sjekk om Maple klarer å faktorisere direkte $2.1x^2 - 3.15xy^3 + y^2x - 1.5y^5$.

Skriv koeffisientene om til rasjonale tall og prøv igjen.

c) Ekspander $(b + 3)^{50}$.

d) Kombiner $\cos(x)\cos(y)$.

e) Forenkl uttrykket $3\cos(2x) - 4\cos(x)^2$.

8. Vi skal kort se på hvordan vi kan bruke Maple til å **løse likninger mhp en variabel**.

For å la Maple løse en **likning** av typen "**uttrykk1** = **uttrykk2**" mhp variabelen x (som da må inngå i minst ett av uttrykkene) kan vi bruke kommandoen

```
[> solve(likning, x);
```

La oss sjekke at Maple kan løse 2.gradslikningen $ax^2 + bx + c = 0$ ved hjelp av "solve":

```
[ > solve(a*x^2+b*x+c=0, x);
```

Og at den kan beregne svaret i et konkret eksempel.

```
[ > solve(x^2+x=1, x);
```

Vi skal videre se på to eksempler der "solve"-kommandoen ikke virker fullt så bra.

Først funderer vi over svaret Maple gir oss på følgende enkle trigonometriske likning:

```
[ > solve(sin(x)-cos(x)=0,x);
```

Det som ikke er "bra" her er at likningen $\sin(x) - \cos(x) = 0$ har uendelige mange løsninger, nemlig alle tall på formen $\frac{\pi}{4} + k\pi$ der k er et helt tall. Kommandoen

"solve" angir her bare en av løsningene, så dette betyr at man kan ikke stole blindt på "solve" (og Maple). Det er ofte lurt i slike tilfeller å bruke Maple til å tegne et grafisk bilde som vi kan bruke til å avgjøre om det kan være flere løsninger enn det "solve" angir. I denne øvelsen trenger vi bare å vite følgende:

Når vi har med en likning i en (reell) variabel x skrevet på formen $f(x) = 0$, kan vi få Maple til å **tegne grafen til $f(x)$ over intervallet $[a, b]$** ved å bruke kommandoen

```
[> plot(f(x), x = a .. b);
```

Merk at det er er nok med **to** prikker mellom a og b ovenfor !

Vi kan da "lese" løsningene av likningen $f(x) = 0$ i intervallet $[a, b]$ ved å finne x -koordinaten til punktene der grafen til $f(x)$ skjærer x -aksen, slik du sikkert har vært vant til å gjøre med en grafisk kalkulator.

La oss nå skissere grafen til $\sin(x) - \cos(x)$ over intervallet $[-2\pi, 2\pi]$:

```
[ > plot(sin(x)-cos(x),x = -2*Pi .. 2*Pi);
```

Vi kan da se at likningen har fire røtter i intervallet $[-2\pi, 2\pi]$. Disse røttene er

$-\frac{7\pi}{4}, -\frac{3\pi}{4}, \frac{\pi}{4}, \frac{5\pi}{4}$, noe som forøvrig kan utledes ved å observere at

$\sin(x) - \cos(x) = 0$ er ekvivalent med at $\sin(x) = \cos(x)$. Funksjonen $\sin(x) - \cos(x)$ er ellers periodisk med periode 2π , så likningen $\sin(x) - \cos(x) = 0$ har to røtter i hver periode og dermed uendelig mange på tallinja.

Legg merke til at vi kan lese av en numerisk approksimasjon av koordinatene til ethvert punkt på grafen ved å peke med muspilen på punktet: koordinatene kommer da opp i et lite vindu på skjermen. **Prøv dette på grafen ovenfor.** Dette kan ofte brukes til å finne en rask approksimasjon av røttene man er ute etter.

La oss nå se på neste eksempel der "solve" ikke kommer direkte med løsningen(e).

```
[ > solve(x^5+2*x+1=0,x);
```

Maple svarer her at løsningene er røttene av den oppgitte femtegradslikning, og det visste vi jo! Problemet er at disse røttene ikke kan skrives på en "pen" form og derfor

må approksimeres. Dette kan vi be Maple om å gjøre ved å bruke følgende kommando

```
[ > allvalues(%);
```

Typografien er ikke helt optimal her, men vi klarer å lese med litt velvilje at likningen har fire komplekse røtter (I står for den imaginære enheten $\sqrt{-1}$) som vi kommer tilbake til senere) og en reell rot (angitt ovenfor med 10 desimaler som tallet - 0.4863890359: Maple sløyfer tallet 0 foran desimalpunktum).

Dette kan vi lett sjekke ved å se på grafen til $x^5 + 2x + 1$ over intervallet $[-1, 1]$.

En kommentar er på sin plass før vi gjør dette: vi bør i slike situasjoner tenke litt over hvilket intervall vi skal velge.

Det hadde ikke vært særlig lurt å velge f.eks. intervallet $[-10, 10]$, noe du gjerne kan prøve. Du vil da se at grafen blir delvis flatklemt på grunn av skalaproblemer og at den er verdiløs for vårt formål. Videre har man også det eventuelle problemet at det kunne finnes røtter utenfor det intervallet man velger: igjen betyr det at man må tenke seg om. Man kan f.eks. prøve å velge et intervall som er slik at vi lett kan begrunne at det ikke finnes røtter utenfor dette intervallet (et slikt intervall vil imidlertid ikke alltid finnes).

I vårt eksempel er det opplagt at $x^5 + 2x + 1$ er større enn 4 når x er større enn 1, og mindre enn -2 når x er mindre enn -1, så eventuelle (reelle) røtter må ligge mellom -1 og 1.

```
[ > plot(x^5+2*x+1, x= -1 .. 1);
```

Dersom vi i utgangspunktet er fornøyd med å finne en **numerisk løsning av en likning** kan vi prøve kommandoen

```
[> fsolve( likning i x , x);
```

i stedet for å bruke "solve"-kommandoen. Med en numerisk løsning menes at løsningen ikke nødvendigvis er en eksakt løsning, men kun en approksimasjon angitt på desimalform. Det er viktig å merke seg at denne kommandoen vil heller ikke alltid finne noen løsning, selv om likningen har løsning(er), og at den vil ikke alltid angi alle løsningene. Videre angir den bare reelle løsninger.

Det er også mulig å angi et intervall $[a, b]$ der Maple skal se etter en mulig løsning ved å utvide kommandoen slik:

```
[> fsolve( likning i x , x, x = a .. b);
```

Det er da ofte lurt å "løse" likningen grafisk først for å se etter hvor eventuelle røtter finnes og derved kunne bestemme seg for valg av intervall. Man bør helst velge "små" intervaller rundt røttene. (Maple's "fsolve"-kommando bruker en variant av Newtons metode, og den metoden er ofte avhengig av et godt valg i utgangspunktet).

Vi prøver "fsolve" på vårt forrige eksempel:

```
[ > fsolve(x^5+2*x+1=0,x);
```

```
[ >
```

Her var det jo bare en reell rot og "fsolve" finner den uten problemer.

Vi prøver deretter "fsolve" på likningen $\sin(x) - \cos(x) = 0$:

```
[ > fsolve(sin(x)-cos(x)=0,x);
```

Svaret likner på $\frac{\pi}{4}$ (slik vi fant med "solve" tidligere) og dette sjekker vi ved

```
[ > evalf(% - Pi/4);
```

```
[ >
```

Vi kan finne løsningen i intervallet $[\pi, 2\pi]$ ved

```
[ > fsolve(sin(x)-cos(x)=0,x,x = Pi .. 2*Pi);
```

som er tilnærmelesvis lik $\frac{5\pi}{4}$ siden

```
[ > evalf(%-5*Pi/4);
```

```
[ >
```

Oppgave 5

a) Betrakt likningen $x^6 - 5x^5 + 2x^4 - x^2 + 5x - 2 = 0$.

Løs denne likningen grafisk , ved hjelp av "solve" og ved hjelp av "fsolve".

b) Betrakt likningen $\cos(2x) = \cos(x)$.

Prøv å løse denne likningen grafisk , ved hjelp av "solve" og ved hjelp av "fsolve".
Gir dette alle løsningene?

9. Noen praktiske kommentarer til slutt.

For å ta vare på arbeidsdokumentet du nå har skrevet i , velger du "**Save as**" i File-rullgardinen. Du må da skrive inn et **navn som avsluttes med .mws**, som f.eks. lab1.mws. Videre må du **angi ditt personlige område** som lagringsområde. Dette dokumentet kan du åpne igjen senere og utføre på nytt.

Når man har brukt %-tegnet i et dokument er det viktig at man utfører alle kommandoene etter tur fra toppen og ned, ellers kan det fort gi gale svar.

Lagrede arbeidsdokumenter får du frem igjen ved å velge "**Open**" i File-rullgardinen og angi dokumentets navn, Du får frem et nytt dokument ved å velge "**New**" samme sted.

For å **avslutte hele programmet** kan du velge "**Quit**" i File-rullgardinen.

DEL 2

1. I del 1 har vi sett at Maple kan håndtere uttrykk i en eller flere variable og at vi kan anvende Maples "subs"-kommandoen til å beregne verdien av et uttrykk ved å fastsette verdien av en eller flere av variablene som inngår i uttrykket. En slik verdifastsettelse av en variabel i "subs"-kommandoen gjelder kun der og da, og Maple betrakter verdien av variabelen som ikke fastlagt i videre beregninger.

Ofte vil det være hensiktsmessig å kunne **fastsette verdien av en variabel**. For eksempel når man skal studere eksponensiell vekst i en variabel x gitt ved et uttrykk på formen $C e^{(kx)}$ vil man gjerne ha behov for å gi variablene C og k forskjellige verdier.

Fastsettelsen av verdien til en variabel gjøres ved **fastsettelseskommandoen**

```
[> variabel := verdi;
```

For eksempel kan vi sette

```
[> k:=3.1;
```

Vi får da at

```
[> (k+1)^2;
```

og

```
[> C*exp(k*x);
```

Legg merke til at bokstaver og navn som ikke er tilordnet noen verdi (slik som C ovenfor) oppfattes som "frie" variable, dvs uten fastlagt verdi.

2. Dersom vi skal utføre mange beregninger er det fort gjort å glemme hvilke variabelnavn vi har brukt tidligere og om disse har fått tilordnet faste verdier. Maple husker faktisk disse verdiene selv om vi går over til et nytt arbeidsdokument og helt til vi avslutter programmet, *med mindre vi velger å klargjøre variabelen(e)*. Med **klargjøring av en variabel** menes at vi opphever det at variabelen har en fastlagt verdi.

Dette gjøres ved **klargjøringskommandoen**

```
[> variabel := ' variabel ' ;
```

alternativt ved kommandoen

```
[> unassign( ' variabel ' );
```

Merk at merke-tegnene før og etter *variabel* **skal** være med i disse kommandoene.

Vi kan altså klargjøre variabelen med navn k som vi brukte ovenfor ved

```
[> k:='k';
```

Vi kan sjekke at k ikke har noen fastlagt verdi ved

```
[ > k;
```

Vi fastlegger så verdien av variabelen C ved

```
[ > C:=-2;
```

Dette gir

```
[ > C*exp(k*x);
```

Vi kan igjen klargjøre C ved

```
[ > unassign('C');
```

Legg merke til at det ikke kommer noe "svar" når vi bruker denne klargjøringskommando.

Vi kan sjekke at C nå ikke har noen fastlagt verdi ved

```
[ > C;
```

3. Vi kan også gi **navn til et uttrykk eller til en likning** ved kommandoene

```
[> navn := uttrykk ;
```

```
[> navn := likning ;
```

Dette er praktisk fordi vi derved slipper å kopiere om og om igjen samme uttrykk/likning dersom vi ønsker å bruke det/den flere ganger. Navn kan vi stort sett velge fritt.

For eksempel kan vi la

```
[ > uttr:=C*exp(k*x);
```

Fastsetter vi nå verdiene av k og C ved

```
[ > k:=2; C:=-1;
```

kan vi sjekke at Maple bruker disse verdiene av k og C når vi ber den om å oppgi hva navnet "uttr" står nå for

```
[ > uttr;
```

Vi kan oppheve fastsettelesen av hva "uttr" står for på samme måte som når vi klargjør variable, altså ved

```
[ > uttr:= 'uttr';
```

eller ved

```
[ > unassign('uttr');
```

La oss se på et eksempel med en likning som vi kaller likn1. Vi setter

```
[ > likn1 := C*exp(k*x)+9*x=1;
```

Legg merke til at Maple oppgir *likn1* med $k = 2$ og $C = -1$ siden vi ikke har klargjort k og C .

Vi kan nå henvise til likningen ovenfor ved å skrive *likn1*. For eksempel kan vi be Maple om å finne en mulig løsning av likningen $-e^{(2x)} + 9x = 1$ i intervallet $[0,1]$ ved

```
[ > fsolve(likn1,x=0..1);
```

Vi kan klargjøre k , C og *likn1* alle på en gang ved å liste dem etter hverandre under klargjøringskommandoen "unassign" på følgende måte

```
[ > unassign('k','C','likn1');
```

Vi sjekker dette ved

```
[ > k;C;likn1;
```

Vi tar med noen nyttige bemerkninger om variabelbruk og Maple.

Navnene x , y , z og t brukes så ofte til å betegne variable at følgende tips kan være nyttig:

fastsett (nesten) aldri verdiene av x , y , z , og t ved fastsettelseskommandoen!

Grunnen er at det er fort gjort å glemme å klargjøre variable, noe som kan føre til at Maple ikke gjør det du forventer.

En måte å **klargjøre alle variablene og alle navn** som har vært brukt er å utføre kommandoen

```
[ > restart;
```

Det lønner seg å alltid begynne et arbeidsdokument med kommandoen "restart". Man vil nemlig ofte ha behov for å gå gjennom alle kommandoene i et dokument på nytt en eller annen gang og det er da best å klargjøre alle variable/navn når man begynner på dokumentet for å unngå feil.

Hvis en har forandret verdien av en variabel flere ganger er det den siste verdien som teller.

Hvis en variabel inngår i definisjonen av et uttrykk er det verdien av variabelen i det øyeblikket definisjonen utføres som teller.

4. For å beregne verdien av et uttrykk i en eller flere variable har vi sett i innføringslaben at vi kan bruke "subs"-kommandoen. Dette kan være tungvint dersom vi ønsker å gjøre det flere ganger og det er da praktisk å definere en funksjon som tilordner verdien av uttrykket til enhver verdi av variabelen (eller variablene) som inngår i formelen for uttrykket. Vi skal bare se på **funksjoner av en variabel** i disse øvelsene (selv om det går an å definere funksjoner av flere variable i Maple).

For å **definere en funksjon av en variabe**, la oss si x , kan vi bruke følgende kommando:

```
[> f := x -> uttrykk (der x inngår) ;
```

Funksjonen $f(x)$ er da definert for alle x som er slik at uttrykket har mening.

Vi kan for eksempel definere $f(x) = \ln(1 + x)$ ved

```
[ > f := x -> ln(1+x);
```

og beregne $f(x)$ for forskjellige verdier av x

```
[ > f(1);  
[ > f(3*a-1);  
[ > 5*f(Pi + 2.7)^2;  
[ > evalf(%);
```

Vi kan også ha definere samme funksjon slik:

```
[ > uttr:=ln(1+x);  
[ > g:=x->uttr;  
[ > g(x);  
[ > f(x)-g(x);
```

Forskjellen mellom det å gi navn til et uttrykk der variabelen x inngår og det å definere en funksjon som tilordner dette uttrykket til hver x er ikke stor. Fra et Maple-synspunkt kan man si at det er hensiktsmessig å definere en funksjon dersom man skal beregne verdien av uttrykket for mange x 'er, for man slipper da å bruke subs-kommandoen mange ganger.

For å skissere **graf** til en funksjon f over intervallet $[a, b]$ bruker vi kommandoen "plot":

```
[> plot( f(x), x = a .. b );
```

For eksempel får vi frem grafen til $\ln(1 + x)$ over intervallet $[-0.5, 4]$ ved

```
[ > plot(f(x), x=-0.5.. 4);
```

Vi kunne ha fått frem samme bildet ved å gi kommandoen $[> \text{plot}(\ln(1+x), x = -0.5..4);$
eller kommandoen $[> \text{plot}(\text{uttr}, x = -0.5 .. 4);$

Maple har ingen problemer med å definere en funksjon av en variabel selv om det inngår andre variable i formelen for uttrykket som beskriver funksjonen.

La oss f. eks. definere funksjonen $h(x) = (10 + 3 \cos(ax) + 4 \sin(ax)) 1.02^x$

```
[ > h:=x->(10 + 3*cos(a*x)+ 4*sin(a*x))* 1.02^x;
```

Beregner vi verdien av $h(x)$ for en verdi av x , f. eks. $x = 3$, vil da svaret bli et uttrykk der a inngår:

```
[ > h(3);
```

La oss nå fastsette verdien av a til å være $\frac{\pi}{2}$ og be Maple tegne grafen over $[0,20]$:

```
[ > a:= Pi/2;  
[ > h(x);  
[ > plot(h(x), x= 0 .. 20);
```

Vi kan gå tilbake til den generelle formelen for $h(x)$ ved å klargjøre variabelen a . Legg merke til at vi da ikke vil kunne skissere grafen til h siden a ikke lenger vil ha en bestemt verdi.

```
[ > a:='a';  
[ > h(x);
```

Kommandoen `[> plot(h(x), x=0..20);` gir nå bare et "tomt" bilde (og ingen feilmelding!). Prøv!

5. Man har ofte bruk for å tegne flere grafer i samme tegning. Vi kan be Maple **tegne to eller flere grafer i samme aksesystem over samme intervall $[a, b]$** ved kommandoen

```
[>plot ( {f(x), g(x), ...}, x=a ..b);
```

La oss illustrere dette ved å skissere grafen til $\sin(x)$ og grafen til x^3 over intervallet $[\frac{\pi}{6}, \frac{\pi}{3}]$:

```
[ > plot({sin(x), x^3}, x=Pi/6..Pi/3);
```

Som vi ser av bildet ovenfor er det all grunn til å tro at grafene skjærer hverandre i et punkt. Dette kan man begrunne formelt ved hjelp av skjæringsetningen. Merk at et skjermbilde aldri kan tolkes som et *bevis* for at to grafer skjærer hverandre, kun som en indikasjon på at dette skjer. For enkelte funksjoner med komplisert graf vil det grafiske bildet på skjermen som Maple angir bare være en dårlig approksimasjon av grafen og et tilsynelatende skjæringspunkt på skjermen trenger ikke å være det i virkeligheten.

6. Maple er ganske effektiv til å beregne **grenseverdier** (dersom de eksisterer). Dette er spesielt nyttig når en skal undersøke såkalte "ubestemte uttrykk" å la " $\frac{0}{0}$ ", " 0∞ ", etc., noe som vi skal lære hvordan man kan håndtere i mange tilfeller ved hjelp av l' Hopitals regel senere i kurset.

Til å beregne **grenseverdien** $\lim_{x \rightarrow a} f(x)$ brukes kommandoen

```
[> limit( f(x), x = a);
```

 dersom a er et tall,

```
[> limit( f(x), x = infinity);
```

 dersom $a = \infty$,

[> **limit(f(x), x = -infinity);** dersom $a = -\infty$.

La oss undersøke grensen $\lim_{x \rightarrow 0} \frac{1 - \cos(x)}{x^2}$.

Direkte innsetting gir et " $\frac{0}{0}$ " -uttrykk som Maple håndterer ved kommandoen

```
[ > limit((1-cos(x))/(x^2), x = 0);
```

At dette svaret ser rimelig ut kan vi sjekke grafisk ved

```
[ > plot((1-cos(x))/(x^2), x=-Pi/2 .. Pi/2);
```

og numerisk ved å definere

```
[ > r:=x->(1-cos(x))/(x^2);
```

og deretter beregne noen av verdiene til r nær 0, f. eks. ved kommandoen

```
[ > [r(-0.1),r(0.01),r(-0.001),r(0.0001)];
```

Det at vi setter klammer rundt verdiene av r listet etter hverandre ovenfor gjør at Maple skriver ut verdiene som en liste. Vær obs. på at Maple angir $r(0.00001) = 0$! Man må da øke antall gjeldende sifre som Maple bruker for å få riktig svar lik 0.5.

7. Maple kan også behandle **ensidige grenser** ved kommandoene

[> **limit(f(x), x=a, right);** for $\lim_{x \rightarrow a^+} f(x)$ og

[> **limit(f(x), x=a, left);** for $\lim_{x \rightarrow a^-} f(x)$.

Vi definerer f.eks. $v(t) = \frac{\sqrt{t}}{|t-1|} - \frac{1}{t-1}$ ved

```
[ > v:=t->(sqrt(t)/abs(t-1)) - 1/(t-1);
```

og beregner de ensidige grenseverdiene av v rundt t=1 ved

```
[ > limit(v(t), t=1, right);
```

```
[ > limit(v(t), t=1, left);
```

Vi kan dermed konkludere at $\lim_{t \rightarrow 1} v(t)$ ikke eksisterer, noe Maple er enig i:

```
[ > limit(v(t), t=1);
```

Vi kan sjekke disse svarene grafisk ved å tegne grafen til v(t) over [1,2] og over [0,1] hver for seg (bildet blir ubrukelig hvis vi istedet bare velger intervallet [0,2])


```
[ > plot(v(t),t=1..2);
```

```
[ > plot(v(t),t=0.5..1,y=0..100);
```

Legg merke til at vi måtte **begrense tillate verdier på y-aksen** ved å tilføye $y = 0 \dots 100$ i plot-kommandoen ovenfor fordi $v(t)$ går mot ∞ når t nærmer seg 1 nedenfra.

Før vi avslutter dette avsnittet om grenseverdier, bør det sies at Maple *ikke* klarer alle grenser. F.eks.

```
[ > limit((1-cos(x))/(x^a),x=0,right);
```

Det er ikke så vanskelig (prøv!) å finne ut for hånd at grenseverdien ovenfor er 0 når $a < 2$, $1/2$ når $a = 2$ og ∞ når $a > 2$. Maple har problemer her med å skille ut de forskjellige tilfellene.

8. Litt mere om graftegning. La oss betrakte funksjonen $f(x) = x + 1 - \sin(x)$

```
[ > f:= x -> x + 1 - sin(x) ;
```

og funksjonen $g(x) = f(x - 2)$

```
[ g:=x->f(x-2); g(x);
```

La oss tegne grafene til disse funksjonene over intervallet $[0, 4\pi]$:

```
[ > plot({f(x),g(x)}, x = 0 .. 4*Pi);
```

Grafen til $f(x)$ er den øverste på skjermen, mens grafen til $g(x)$ er den nederste. Vi ser at en del av grafen til $g(x)$ fremkommer ved en horisontal forskyvning mot høyre av en del av grafen til $f(x)$ med to enheter.

Nå ba vi om at begge grafene ble tegnet over samme intervall $[0, 4\pi]$. Hvis vi virkelig vil ha et bilde av hele den forskjøvne graf, må grafen til $g(x) = f(x-2)$ tegnes over intervallet $[2, 2 + 4\pi]$.

Generelt er grafen til $g(x) = f(x-c)$ over intervallet $[a+c, b+c]$ en horisontal forskyvning av grafen til $f(x)$ over $[a,b]$ med $|c|$ enheter, mot høyre hvis $c > 0$ og mot venstre hvis $c < 0$. I tilfellet ovenfor er $c = 2 > 0$ og $|c| = 2$.

For å kunne vise dette i samme tegning, må vi bruke kommandoen "**display**", som først blir tilgjengelig ved å laste ned **tilleggspakken "with(plots)"**. Dette gjøres ved å utføre kommandoen

```
[ > with(plots):
```

Legg merke til at vi brukte **kolon** i kommandoen ovenfor. Dette er for å unngå at Maple lister ut alle de mulighetene som tilleggspakken inneholder.

Når denne tilleggspakken er lastet ned kan vi definere grafene til to eller flere

funksjoner over muligens forskjellige intervaller og be Maple skissere alle grafene i samme aksesystem ved kommandoen

```
[> display( graf1, graf2, ... )];
```

For eksempel kan vi nå få med hele forskyvningen av grafen til $f(x)$ over $[0, 2\pi]$ med 2 enheter mot høyre ved

```
[ > graf1:=plot(f(x),x=0..4*Pi):  
  graf2:=plot(g(x),x=2..2+4*Pi):  
[ > display({graf1,graf2});
```

Det går an å velge farger for hver graf hvis man vil. På skjermen din er begge grafene ovenfor gjengitt i samme farge og det kan i andre tilfeller være vanskelig å skille grafene fra hverandre når man bruker display-kommandoen slik vi gjorde her.

9. Derivasjon er ingen kunst og Maple deriverer i vei som bare det. Vi kan bruke to alternative måter avhengig av om vi ønsker å derivere et uttrykk eller en funksjon.

- Maple angir den **deriverte mhp x av et (deriverbart) uttrykk i variabelen x** ved kommandoen

```
[> diff( uttrykk , x);
```

- Hvis f angir en (deriverbar) funksjon av x , kan vi be Maple om å angi **den deriverte $f'(x)$** ved

```
[> D(f)(x);
```

La oss f. eks. betrakte funksjonen $f(x) = -6 \left(\sqrt{3} \cos\left(\frac{\pi x}{6}\right) + \sin\left(\frac{\pi x}{6}\right) \right)$:

```
[ > f:=x->-6*(sqrt(3)*cos(Pi/6*x) + sin(Pi/6*x));
```

Vi bruker begge måtene til å finne et uttrykk for $f'(x)$ og beregne $f'(2)$:

```
[ > uttr:=diff(f(x),x);  
[ > subs(x=2,uttr);  
[ > simplify(%);
```

```
[ > D(f)(x);  
[ > D(f)(2);
```

Merk at vi kan bruke $D(f)(x)$ bare når funksjonen $f(x)$ først er definert.

Per definisjon av den deriverte vet vi at $f'(2) = \lim_{h \rightarrow 0} \frac{f(2+h) - f(2)}{h}$,

og generelt at $f'(x) = \lim_{h \rightarrow 0} \frac{f(x+h) - f(x)}{h}$.

Dette kan vi lett sjekke ved

```
[ > limit( (f(2+h)-f(2))/h,h=0);
```

```
[ > limit( (f(x+h)-f(x))/h,h=0);
```

som gir de samme svarene som ovenfor.

Maple "kan" regnereglerne for derivasjon. F.eks. kan vi gjenfinne derivasjonsregelen for et produkt ved

```
[ > diff( h(x)*k(x),x);
```

Merk at Maple skriver her $\frac{\partial}{\partial x} h(x)$ istedet for $h'(x) = \frac{d}{dx} h(x)$.

10. La oss igjen betrakte funksjonen $f(x)$ som vi definerte ovenfor.

Vi skal illustrere den geometriske betydningen av $f'(2) = \pi$ som vi vet kan tolkes som stigningstallet til tangenten til grafen til $f(x)$ i punktet $(2, f(2))$.

Vi gjør først en liten digresjon og minner om at en linje i planet med stigningstall m som går gjennom punktet (a, b) er grafen til funksjonen $l(x) = m(x - a) + b$.

Vi setter derfor

```
[ > l:=x-> m*(x-a)+b;
```

Vi kan da få frem tangenten $t(x)$ til grafen til $f(x)$ i punktet $(a, b) = (2, f(2))$ ved å sette

```
[ > a:=2; b:=f(2); m:=D(f)(2);
```

og deretter

```
[ > tangent:=l(x); t:=x->tangent;  
[ > t(x);
```

Merk at vi **ikke** definerte direkte tangenten $t(x)$ ved å sette $t:=x \rightarrow l(x)$;

Grunnen til det er at funksjonen $t(x)$ da ville bli lik funksjonen $l(x)$. Siden variablene a , b og m inngår i selve definisjonen av $l(x)$ ville da $t(x)$ forandre seg hver gang vi forandret en av verdiene av a , b , eller m og det er ikke ønskelig her.

La oss også betrakte linjen (sekanten) som går gjennom punktene $(a, b) = (2, f(2))$ og $(3, f(3))$.

Denne linjen har stigningstall $m = \frac{f(3) - f(2)}{3 - 2} = f(3) - f(2)$, så den er grafen til $s(x)$

der $s(x)$ kan bestemmes ved hjelp av $l(x)$ ved å først sette

```
[ > m:=f(3)-f(2);
```

Deretter setter vi (husk her at a fremdeles er fastsatt til 2 og b til $f(2)$)

```
[ > sekant:=1(x); s:=x->sekant;
[ > s(x);
```

Vi kan så tegne grafene til $f(x)$, $t(x)$ og $s(x)$ i samme aksesystem over intervallet $[0, 6]$ ved

```
[ > plot({f(x),t(x),s(x)},x=0 .. 6);
```

I plot-kommandoen ovenfor kunne vi godt ha brukt "tangent" istedet for "t(x)" og "sekant" istedet for "s(x)", så det var strengt tatt unødvendig å definere $t(x)$ og $s(x)$ her. Dette var hovedsaklig ment som en trening i å definere nye funksjoner.

DEL 3

Maple er rimelig god til å finne ubestemte integraler (m.a.o. til å antiderivere) dersom disse lar seg uttrykke ved hjelp av elementære funksjoner. På den andre siden er den veldig god til å beregne bestemte integraler numerisk.

1. For å beregne **det ubestemte integralet** $\int f(x) dx$ brukes kommandoen

```
[> int ( f(x) , x ) ;
```

Når Maple ikke klarer å finne en antiderivert av $f(x)$ returneres bare $\int f(x) dx$.

Ellers oppgir Maple bare *en* antiderivert for $f(x)$: den tar *ikke* med en integrasjonskonstant i svaret.

La oss f. eks beregne $\int \frac{x^2(2-x)}{x^4+1} dx$ (som er ganske tidkrevende å beregne for hånd):

```
[ > f:=x->x^2*(2-x)/(x^4+1);
```

$$f := x \rightarrow \frac{x^2(2-x)}{x^4+1}$$

```
[ > int(f(x),x);
```

$$\frac{1}{4}\sqrt{2} \ln\left(\frac{x^2 - x\sqrt{2} + 1}{x^2 + x\sqrt{2} + 1}\right) + \frac{1}{2}\sqrt{2} \arctan(x\sqrt{2} + 1) + \frac{1}{2}\sqrt{2} \arctan(x\sqrt{2} - 1) - \frac{1}{4} \ln(x^4 + 1)$$

Vi bør jo stole på at Maple regner riktig, men la oss sjekke at dette svaret er virkelig

en antiderivert av $f(x) = \frac{x^2(2-x)}{x^4+1}$:

```
[ > diff(%,x);
```

$$\left[\frac{1}{4} \frac{\sqrt{2} \left(\frac{2x - \sqrt{2}}{x^2 + x\sqrt{2} + 1} - \frac{(x^2 - x\sqrt{2} + 1)(2x + \sqrt{2})}{(x^2 + x\sqrt{2} + 1)^2} \right) (x^2 + x\sqrt{2} + 1)}{x^2 - x\sqrt{2} + 1} + \frac{1}{1 + (x\sqrt{2} + 1)^2} \right. \\ \left. + \frac{1}{1 + (x\sqrt{2} - 1)^2} - \frac{x^3}{x^4 + 1} \right]$$

Dette svaret likner ikke helt på $f(x)$, så vi omformer det:

$$\left[> \text{factor}(\%); \right. \\ \left. - \frac{(-2 + x)x^2}{(x^2 + x\sqrt{2} + 1)(x^2 - x\sqrt{2} + 1)} \right]$$

At uttrykket ovenfor er det samme som $f(x)$ følger av at dets nevner er det samme som $x^4 + 1$. Dette kan vi sjekke ved

$$\left[> \text{expand}(\text{denom}(\%)); \quad \# \text{ OBS: denom}(\%) \text{ angir nevneren i } \% ! \\ x^4 + 1 \right]$$

I dette eksemplet hadde vi ikke trengt å definere funksjonen $f(x)$ først : det hadde vært nok å bruke uttrykket som beskriver $f(x)$.

La oss nå be Maple om å prøve å beregne $\int \sin(x^2(2-x)) dx$:

$$\left[> \text{int}(\sin(x^2*(2-x)), x); \right. \\ \left. \int \sin(x^2(2-x)) dx \right]$$

Legg merke til hvor (relativt) lang tid Maple bruker før den svarer ved å returnere integralet tilbake: den prøver gjennom hele sitt repertoar før den gir opp!

2. Til å beregne **det bestemte integralet** $\int_a^b f(x) dx$ brukes kommandoen

$$[> \text{int}(f(x), x = a .. b);$$

F. eks. kan vi beregne $\int_0^2 \frac{x^2(2-x)}{x^4+1} dx$ ved

$$\left[> \text{int}(x^2*(2-x)/(x^4+1), x = 0..2); \right]$$

$$\left[\begin{aligned} & \frac{1}{4}\sqrt{2} \ln(5 - 2\sqrt{2}) - \frac{1}{4}\sqrt{2} \ln(5 + 2\sqrt{2}) + \frac{1}{2}\sqrt{2} \arctan(2\sqrt{2} + 1) + \frac{1}{2}\sqrt{2} \arctan(2\sqrt{2} - 1) \\ & - \frac{1}{4}\ln(17) \end{aligned} \right.$$

og vi kan da få svaret ut på desimalform ved

```
[ > evalf(%);
                                     .5252220310
```

Vi kunne også ha skrevet direkte

```
[ > evalf(int(x^2*(2-x)/(x^4+1), x = 0..2));
                                     .5252220310
```

eller vi kunne ha oppgitt en av integrasjonsgrensene på desimalform:

```
[ > int(x^2*(2-x)/(x^4+1), x = 0..2.0);
                                     .5252220310
```

La oss prøve å beregne $\int_0^2 \sin(x^2(2-x)) dx$ ved disse tre alternative måter:

```
[ > int(sin(x^2*(2-x)), x = 0..2);
                                      $\int_0^2 \sin(x^2(2-x)) dx$ 
```

```
[ > evalf(int(sin(x^2*(2-x)), x = 0..2));
                                     1.141205400
```

```
[ > int(sin(x^2*(2-x)), x = 0..2.0);
                                      $\int_0^{2.0} \sin(x^2(2-x)) dx$ 
```

I første og tredje alternativ forsøker Maple å finne en antiderivert, gir opp, og returner bare det bestemte integralet tilbake. I det andre alternativet returnerer Maple et desimaltall som svar. Dette er fordi Maple da velger å finne et *approximativt* svar, og den bruker da en avansert metode for såkalt numerisk integrasjon.

Vi oppsummerer dette slik:

Til å beregne $\int_a^b f(x) dx$ ved **numerisk integrasjon** brukes kommandoen

```
[> evalf( int( f(x), x = a .. b));
```

La oss beregne $A = \int_1^4 \ln(x) dx$ på desimalform ved å bruke begge måtene:

```
[ > A := int(ln(x), x= 1..4.0);
                                     A = 2.545177444
[ > A = evalf(int(ln(x), x= 1..4));
                                     I = 2.545177445
```

Som vi ser er det en minimal forskjell mellom disse to svarene. Det andre svaret beregnes direkte ved numerisk integrasjon. Under utførelsen av den første kommandoen beregner Maple først en antiderivert $F(x)$ av $f(x) = \ln(x)$ (som du lett kan beregne selv ved delvis integrasjon). Deretter bruker Maple analysens

fundamentalsats, som her sier at $\int_1^{4.0} \ln(x) dx = F(4.0) - F(1)$, og den regner til slutt dette svaret om til desimalform. Vi kunne ha utført trinnene i denne prosessen slik:

```
[ > F:=int(ln(x),x);
                                     F := x ln(x) - x
[ > subs(x=4.0,F)-subs(x=1,F);
                                     4.0 ln(4.0) - 3.0 - ln(1)
[ > A = evalf(%);
                                     A = 2.545177444
```

Vi kan forøvrig lett sjekke her at det oppgitte uttrykket F er en antideriverte av $\ln(x)$ ved

```
[ > diff(F,x);
                                     ln(x)
```

I Maples studentpakke "**student**", som du kan laste ned ved å utføre kommandoen [**> with(student):**], finnes det bl.a. egne kommandoer for å utføre **substitusjon** eller **delvis integrasjon** i integraler.

Oppgave 1

a) Finn $\int \frac{\sqrt{x} + 1}{1 - \sqrt{x}} dx$ og $A = \int_4^9 \frac{\sqrt{x} + 1}{1 - \sqrt{x}} dx$ ved hjelp av Maple.

b) Finn $\int \frac{\tan(v)}{1 + \cos(v)} dv$ og $B = \int_0^{\frac{\pi}{3}} \frac{\tan(v)}{1 + \cos(v)} dv$ ved hjelp av Maple.

c) Angi A og B på desimalform ved å bruke tre forskjellige måter.

3. I dette kapittelet skal vi se på hvordan Maple kan brukes til å studere *ordinære* differensial likninger (forkortes til ODE). En ODE er en likning som involverer en deriverbar funksjon av en variabel x , definert på et passende intervall, dens deriverte og/eller eventuelt noen av dens høyere ordens deriverte. Den høyeste orden av alle de deriverte som inngår i likningen kalles *ordenen* av likningen.

I MAT 1010 studerer vi bare spesielle ODE' er av 1. eller 2. orden. Maple kan løse eksakt mange slike likninger. Dersom Maple ikke klarer å løse en ODE eksakt kan man be den finne en numerisk løsning.

For ordens skyld begynner vi med Maples kommandoer for å behandle ODE' er. For å be Maple om å forsøke å finne løsningene av en ODE som involverer en funksjon $y(x)$ av x , brukes kommandoen.

`[> dsolve("ODE for y(x)", y(x));`

Man **må** her skrive $y(x)$ og ikke bare y , også når man skriver inn hva ODE skal være.

Dersom Maple klarer å angi løsningene på eksakt form vil det angi disse ved hjelp av det nødvendige antall konstanter som Maple betegner med `_C1`, `_C2`, osv... Antall konstanter avhenger av ordenen på ODE' en.

Legg merke til at det her ikke angies noen intervall for løsningene. Grunnen er at forskjellige løsninger av samme ODE ofte er definert på forskjellige intervaller. Er man interessert i løsninger på et bestemt intervall må man velge ut hvilke av løsningene som er definert akkurat på det intervallet.

Dersom Maple ikke klarer jobben, returneres ingenting. Dette vil generelt skje ofte: f. eks. er det å løse den enkle 1. ordens ODE' en $y' = f(x)$ det samme som å finne en antiderivert av $f(x)$ og det lar seg ofte ikke gjøre på eksakt form.

La oss se på noen eksempler:

a) Den 1. ordens lineære ODE $y' - 2xy = x^3$ løses lett ved

```
[ > dlikna := diff(y(x),x) - (2*x)*y(x) = x^3;
```

$$dlikna := \left(\frac{\partial}{\partial x} y(x) \right) - 2xy(x) = x^3$$

(Vi kan godt skrive ODE'en rett inn i dsolve, men det blir mere oversiktlig når man gir

den et navn.)

```
> dsolve(dlikna,y(x));
```

$$y(x) = -\frac{1}{2}x^2 - \frac{1}{2} + e^{(x^2)} _C1$$

Vi ser at det inngår en konstant $_C1$ i svaret og at løsningen er definert for alle x , uansett verdien av $_C1$.

Fra teorien vet man at grafene til løsningene som svarer til forskjellige valg av $_C1$ aldri skjærer hverandre. For å illustrere dette er det hensiktsmessig å gi et navn til uttrykket som beskriver $y(x)$ i svaret ovenfor. Dette uttrykket kommer i på **høyre** siden av likhetstegnet i forrige Maple svar og vi kan henvise til det ved "**rhs(%)**": Maple tolker nemlig **rhs** som "**right hand side of**". Vi setter derfor

```
> sol:=rhs(%);
```

$$sol := -\frac{1}{2}x^2 - \frac{1}{2} + e^{(x^2)} _C1$$

Vi velger nå $_C1$ til å være -1, 0 og 1 etter tur og tegner grafen til disse tre løsningene på (0, 1) (ikke tegnet her, men lett å få fram ved å endre : til ;):

```
> asolm1:= subs(_C1=-1,sol);
asol0:= subs(_C1=0,sol);
asol1:= subs(_C1=1,sol);
plot({asolm1,asol0,asol1}, x= -1..1):
```

$$asolm1 := -\frac{1}{2}x^2 - \frac{1}{2} - e^{(x^2)}$$
$$asol0 := -\frac{1}{2}x^2 - \frac{1}{2}$$
$$asol1 := -\frac{1}{2}x^2 - \frac{1}{2} + e^{(x^2)}$$

En liten bemerkning : når vi skrev inn ODE'en, brukte vi **diff(y(x), x)** til å betegne $y'(x)$. Vi kunne også ha skrevet inn **D(y)(x)** i stedet.

b) Den (separable) 1. ordens ODE'en $(x + 1) y' + y^2 = 0$ "løses" ved

```
> dliknb:=(x+1)*D(y)(x) + y(x)^2 = 0;
```

$$dliknb := (x + 1) D(y)(x) + y(x)^2 = 0$$

```
> dsolve(dliknb,y(x));
```

$$y(x) = \frac{1}{\ln(x + 1) + _C1}$$

Her ser vi et lite problem med Maple: den betrakter den naturlige logaritmen som en (kompleks) funksjon definert for (de fleste) komplekse tall og den inkluderer derfor ikke tallverdien innenfor "ln" i svaret ovenfor. Regner du denne oppgaven for hånd vil du komme frem til at det bør stå $\ln|x+1|$ i svaret ovenfor. Dette spiller jo ingen rolle dersom $x > -1$, men det er ingen grunn her til at vi ikke også skulle være interessert i tilfellet $x < -1$.

Man bør derfor alltid tenke over svaret som Maple gir dersom det inngår logaritmer i det!

Legg videre merke til at intervallene der $y(x)$ er definert i svaret ovenfor avhenger i høy grad av konstanten $_C1$: $\ln|x+1| + _C1$ må være forskjellig fra 0 samtidig som x må være forskjellig fra -1 for at det (korrekte) uttrykket for $y(x)$ skal ha en mening.

Til å betegne $y''(x)$ kan vi bruke enten **diff(y(x), x\$2)** eller **D(D(y))(x)**. La oss bruke den siste skrivemåten i neste eksempel:

c) Den 2. ordens lineære ODE'en $y'' - 2y' - 8y = 6 - 8x$ løses uten problemer:

```
[ > dliknc:= D(D(y))(x) - 2*D(y)(x) - 8* y(x) = 6 - 8*x;
      dliknc := (D(2))(y)(x) - 2 D(y)(x) - 8 y(x) = 6 - 8 x
[ > dsolve(dliknc,y(x));
      y(x) = -1 + x + _C1 e(4x) + _C2 e(-2x)
```

Her ser vi at inngår det som ventet to konstanter i svaret.

d) Følgende 1.ordens ODE : $y' = \sin(y) + x$ er ikke lett (umulig?) å løse eksakt:

```
[ > dliknd:= D(y)(x)= sin(y(x)) + x;
      dliknd := D(y)(x) = sin(y(x)) + x
[ > dsolve( dliknd, y(x));
[ >
```

Maple gir intet svar her (og vi klarer heller ikke å løse denne eksakt ?!).

Oppgave 2:

Løs følgende ODE' er ved hjelp av Maple:

i) $y' - 2xy = e^{(x^2)}$. Skisser (minst) tre av løsningene på intervallet (-1,1).

ii) $(x^2 + 1) y' + y^2 + 1 = 0$.

iii) $y'' + 3y' + 2y = 4x + 10 \sin(x)$.

4. Konstantene som inngår i den (eventuelle) generelle løsningen av en ODE vil som regel kunne bestemmes dersom det oppgis nok tilleggskrav slik at den søkte løsningen er entydig. Slike tilleggskrav gjelder vanligvis verdien(e) av y (og evt. noen av dens deriverte) for en bestemt verdi av x (kravet/kravene kalles da "**initialbetingelse(r)**"), eller for to eller flere oppgitte verdier av x (kravene kalles da ofte for "**randbetingelser**"). Maples dsolve-kommando kan brukes med slike tilleggskrav på følgende måte:

```
[> dsolve( {"ODE for y(x)", "tilleggskrav"}, y(x) );
```

Et tilleggskrav av typen $y(\mathbf{a}) = \mathbf{b}$ skrives inn **rett frem**, mens et tilleggskrav av typen $y'(\mathbf{a}) = \mathbf{c}$ må skrives inn som $\mathbf{D}(y)(\mathbf{a}) = \mathbf{c}$. Klarer ikke "dsolve" jobben returneres intet svar. Dette gjelder også dersom det er for mange motstridende tilleggskrav.

Vi ser på de samme eksemplene som før, men nå med tilleggskrav. Merk at vi bruker for enkelhets skyld navnene på diff. likningene vi innførte tidligere!

a) ODE'en $y' - 2xy = x^3$ med tilleggskravet $y(0)=1$ løses ved

```
[ > dsolve({dlikna, y(0)=1}, y(x));
```

$$y(x) = -\frac{1}{2}x^2 - \frac{1}{2} + \frac{3}{2}e^{x^2}$$

b) ODE'en $(x + 1) y' + y^2 = 0$ med tilleggskravet $y(0) = 1/2$ forsøkes løst på $(-1, \infty)$ ved

```
[ > dsolve( {dliknb, y(0) = 1/2}, y(x) );
```

$$y(x) = \frac{1}{\ln(x + 1) + 2}$$

Vi må passe på litt her fordi denne løsningen er ikke definert på hele $(-1, \infty)$. Dette kan vi se slik:

```
[ > sol:=rhs(%);
```

$$sol := \frac{1}{\ln(x + 1) + 2}$$

```
[ > u:=denom(sol);
```

$$u := \ln(x + 1) + 2$$

```
[ > a:=solve(u=0,x);
```

$$a := e^{(-2)} - 1$$

```
[ > evalf(a);
                                     -.8646647168
```

Dette betyr at løsningen ovenfor bare er definert på $(-1,a)$ og på (a, ∞) , der a er som ovenfor. Siden vi har brukt verdien av $y(x)$ i $x = 0$ til å bestemme $y(x)$, og siden 0 ligger i intervallet (a, ∞) , er det naturlig å si at løsningen vi fant gjelder i intervallet (a, ∞) . Dette er opplagt riktig dersom vi bare er interessert i en løsning som gjelder i det største intervallet som inneholder 0 .

En skisse av løsningen på intervallet $(a,2)$ får en fram ved kommandoen :

```
[ > plot( sol,x = a..2 , y = 0..5):
```

Vi begrenset her verdiene av y til å ligge mellom 0 og 5 fordi $x=a$ er en vertikal asymptote for $y(x)$. Du kan gjerne prøve med andre begrensninger av y -verdiene for å se hvordan bildet forandrer seg. Det må gjerne litt prøving og feiling til før man kommer til et bilde man kan si seg fornøyd med.

c) ODE'en $y'' - 2y' - 8y = 6 - 8x$ med tilleggskravene $y(1)=y'(1) = 1$ løses slik:

```
[ > dsolve( {dliknc, y(1)=1, D(y)(1)=1}, y(x));
```

$$y(x) = -1 + x + \frac{1}{3} \frac{e^{(4x)}}{(e^2)^2} + \frac{2}{3} e^2 e^{(-2x)}$$

```
[ > simplify(%);
```

$$y(x) = \frac{1}{3} (-3 e^{(4-4x)} + 3 x e^{(4-4x)} + 1 + 2 e^{(6-6x)}) e^{(-4+4x)}$$

Vi kan få et grafisk inntrykk av løsningen rundt punktet $(1,1)$ ved å bruke kommandoen

```
[ > plot(rhs(%),x=0..1.5, scaling=constrained):
```

Vi bruker her samme skala på begge aksene slik at vi kan "se" at $y'(1) = 1$.

c') ODE'en $y'' - 2y' - 8y = 6 - 8x$ med "randbetingelsene" $y(0) = 2, y(1) = 1$ løses slik:

```
[ > dsolve( {dliknc, y(0)=2,y(1)=1}, y(x));
```

$$y(x) = -1 + x + \frac{(3 e^{(-2)} - 1) e^{(4x)}}{-e^4 + e^{(-2)}} - \frac{(3 e^4 - 1) e^{(-2x)}}{-e^4 + e^{(-2)}}$$

c'') ODE'en $y'' - 2y' - 8y = 6 - 8x$ med de **tre** tilleggskravene

$$y(0) = 2, y(1) = 1, y'(1) = 1$$

har ingen løsninger:

```
[ > dsolve( {dliknc, y(0)=2,y(1)=1,D(y)(1)=1}, y(x));
[ >
```

Det betyr spesielt at den entydige løsningen vi fant under punktet c') ikke tilfredstiller at $y'(1) = 1$.

c''') Det går an å angi færre tilleggskrav enn det som er nødvendig for å bestemme løsningen entydig. Svaret (hvis den kommer!) inneholder da en eller flere ubestemte konstanter. Et slikt eksempel er:

```
[ > dsolve( {dliknc, y(1)=1}, y(x));
[
      
$$y(x) = -1 + x + \_C1 e^{(4x)} + (-\_C1 (e^2)^3 + e^2) e^{(-2x)}$$

```

d) ODE'en $y' = \sin(y) + x$ med tilleggskravet $y(0) = 2$ lar seg fremdeles ikke løse eksakt:

```
[ > dsolve( {dliknd, y(0)=2}, y(x));
[ >
```

Vi kan løse dette problemet numerisk ved hjelp av Maple, men dette venter vi med til slutt.

Oppgave 3

Løs følgende ODE' er med tilleggskrav ved hjelp av Maple:

i) $y' - 2x y = e^{(x^2)}, y(0) = 0.$

ii) $(x^2 + 1) y' + y^2 + 1 = 0, y(0) = 2.$

iii) $y'' + 3 y' + 2 y = 4x + 10 \sin(x), y(0) = y'(0) = 1.$

iv) $y'' + 3 y' + 2 y = 4x + 10 \sin(x), y(0) = y(1) = 0.$

5. Maple klarer å løse enkelte ODE' er der det inngår ubestemte konstanter. La oss f.eks. betrakte en homogen lineær 2. ordens ODE med konstante koeffisienter av følgende type:

$$y'' + q y' + y = 0 \quad (\text{der } q \text{ er en reell konstant}),$$

med følgende initialbetingelser : $y(0) = -1, y'(0) = 0$.

Denne kan jo lett løses for hånd. Svarets utseende vil da avhenge av verdien av parameteren q . Til orientering nevner vi at denne ODE'en kan tolkes som en forenklet, kvalitativ modell for en "svingende" kloss, der konstanten q representerer friksjonen i systemet. La oss se hva Maple gjør med denne ODE:

```
[ > dliknf:= D(D(y))(x) + q*D(y)(x) + y(x) =0 ;
      dliknf:= (D^(2))(y)(x) + q D(y)(x) + y(x) = 0
```

```
[ > dsolve({dliknf, y(0)=-1, D(y)(0)=1}, y(x));
```

$$y(x) = -\frac{1}{2} \frac{(q + \sqrt{q^2 - 4} - 2) e^{(-1/2 (q - \sqrt{(q-2)(q+2)})x}}{\sqrt{q^2 - 4}} + \frac{1}{2} \frac{(q - \sqrt{q^2 - 4} - 2) e^{(-1/2 (q + \sqrt{(q-2)(q+2)})x}}{\sqrt{q^2 - 4}}$$

```
[ > solq:=rhs(%);
```

$$solq := -\frac{1}{2} \frac{(q + \sqrt{q^2 - 4} - 2) e^{(-1/2 (q - \sqrt{(q-2)(q+2)})x}}{\sqrt{q^2 - 4}} + \frac{1}{2} \frac{(q - \sqrt{q^2 - 4} - 2) e^{(-1/2 (q + \sqrt{(q-2)(q+2)})x}}{\sqrt{q^2 - 4}}$$

Det er flere ting å bemerke ved dette svaret. Først ser vi at $q^2 - 4$ opptrer i nevneren, og det betyr at Maple overser tilfellene $q = 2$ og $q = -2$. Disse bør da studeres for seg. Det neste er at svaret ikke likner helt på hva vi kunne ha forventet: det skiller ikke mellom verdier av q slik vi hadde gjort for hånd. Grunnen er at Maple bruker **komplekse** eksponentialfunksjoner i sitt svar: leddet $\sqrt{(q-2)(q+2)}$ må tolkes som en kompleks kvadratroten av $(q-2)(q+2)$ dersom $|q| < 2$. Svaret ovenfor *kan* skrives om på reell form ved å se på tilfellene $|q| < 2$ og $2 < |q|$ hver for seg!

La oss illustrere dette ved å sette $q = 0$ ("udempet kloss"), $q = 0.5$ ("dempet kloss") og $q = 3$ ("overkritisk dempet kloss") etter tur.

```
[ > q:=0; solq;
```

$$q := 0$$

$$\frac{1}{8} (-2 + \sqrt{-4}) \sqrt{-4} e^{(1/2 \sqrt{-4} x)} - \frac{1}{8} (-2 - \sqrt{-4}) \sqrt{-4} e^{(-1/2 \sqrt{-4} x)}$$

Til å gjøre om dette uttrykket på reell form brukes kommandoen [$>$ evalc("uttrykk");

```
[ > evalc(%);
```

```

[ -cos(1/2*sqrt(4)*x) + 1/2*sqrt(4)*sin(1/2*sqrt(4)*x)
> solq0:=simplify(%);
  solq0 := -cos(x) + sin(x)

> q:=0.5; solq;
  q := .5
(-.5000000000 - .3872983347 I) e^{((-2500000000 + .9682458365 I) x)}
+ (-.5000000000 + .3872983347 I) e^{((-2500000000 - .9682458365 I) x)}

> solq1:=evalc(%);
solq1 := -1.0000000000 e^{(-.2500000000 x)} cos(.9682458365 x)
+ .7745966694 e^{(-.2500000000 x)} sin(.9682458365 x)

> q:=3; solq2:=solq;
  q := 3
solq2 := -1/10*(sqrt(5)+1)*sqrt(5)*e^{(-1/2*(3-sqrt(5))*x)} + 1/10*(1-sqrt(5))*sqrt(5)*e^{(-1/2*(3+sqrt(5))*x)}

```

Vi kan nå skissere grafene til disse tre løsningene ved den neste kommandoen. Du bør klare å skjønne hvilke graf som hører til de forskjellige verdiene av q:

```

[ > plot({solq0,solq1,solq2}, x=0.. 10):

```

Oppgave 4:

Studer hva som skjer i eksemplet ovenfor når du setter $q = -0.5$ og $q = -3$ etter tur. Lag en skisse av grafene til løsningene over intervallet $[0, 10]$ (ikke gjør det i samme tegning pga skaleringsproblemer).

(Kommentar: det at q er negativ kan tolkes som "negativ friksjon", altså at klossen får positiv "drahjelp" hele tiden!).

Oppgave 5:

Betrakt ODE 'en $y'' + y = \sin(p x)$ med initialbetingelsene $y(0) = -1, y'(0) = 0$.

(Dette er en forenklet modell for en svingende kloss, uten friksjon, men utsatt for ytre regelmessige vibrasjoner).

Bruk dsolve til å løse denne og skisser løsningene for $p = 1/2, p = 1$ og $p = 2$. (Tilfellet $p=1$ må behandles separat).

Hva skjer med klossen når (tiden) x går mot uendelig i hvert tilfelle ?

6. Vi skal se "geometrisk" på 1. ordens ODE'er av typen

$$(*) \quad y' = f(x, y)$$

der $f(x, y)$ er et uttrykk i x og y .

La oss betrakte en løsning $y(x)$ av (*) hvis graf går gjennom et punkt (x_0, y_0) . Det siste betyr at $y_0 = y(x_0)$.

Tangenten L til grafen til $y(x)$ i dette punktet har stigningstall $y'(x_0) = f(x_0, y(x_0)) = f(x_0, y_0)$.

En retningsvektor for linjen L er derfor gitt ved vektoren $(1, f(x_0, y_0))$.

Vi kan tenke oss at denne vektoren angir "kompassretningen" i (x_0, y_0) for vår videre vei dersom vi går langs grafen til $y(x)$ (fra venstre mot høyre). Tegner vi flere av vektorene på formen $(1, f(x, y))$, hver av disse plassert i det tilhørende punktet (x, y) , får vi frem et "felt" av retningsvektorer som må tangeres av løsningskurvene til (*).

En illustrasjon av et slikt felt i et rektangel i xy -planet kan vi få Maple til å tegne ved kommandoen

```
[> fieldplot ([1, f(x,y)], x= a..b, y=c..d);
```

Merk at denne kommandoen først er tilgjengelig når man laster inn pakken **with(plots)**.

Som et eksempel betrakter vi ODE'en $y' - 2xy = x^3$ som vi studerte først i denne øvelsen.

Denne kan jo skrives på formen $y' = 2xy + x^3$, som er et uttrykk i x og y .

Vi prøver "fieldplot" kommandoen på denne, og tar med grafen til tre av løsningene vi fant tidligere. Vi må derfor bruke kommandoen "display" for å få dette til.

Legg videre merke til at vi avslutter de tre første kommandoene med kolon for å unngå å få masse unødvendige ting frem på skjermen! Sett semi-kolon etter den siste kommandoen for å få fram bildet.

```
[> with(plots):  
bilde1:=fieldplot([1,(2*x)*y + x^3], x=-1..1,y=-4..2):  
bilde2:=plot({asolm1,asol0,asol1}, x=-1..1):  
display({bilde1,bilde2}):
```


Slike bilder av retningsvektorfelder er ikke alltid gode fordi mange av vektorene knapt er synlige, mest fordi disse blir skalert ned. Disse bildene kan f.eks tenkes å beskrive strømretningene på overflaten av en elv, der en partikkel som slippes ned i elven vil følge en kurve bestemt av disse. Som vi også ser av bildet, så kan veiene bli ganske forskjellige selv om utgangspunktene er forholdsvis nær hverandre.

Bilder av retningsvektorfeltet for en ODE som (*) er spesielt nyttige når man ikke klarer å løse den eksakt. De gir jo et vist inntrykk av hvordan løsningskurvene må se ut.

La oss prøve dette på likningen $y' = \sin(y) + x$ (som vi ikke klarer å løse eksakt):

```
[ > bilde3:=fieldplot([1,sin(y)+x], x=-1.5..1.5, y=0..4):
  display(bilde3):
```

Vi kan få frem en løsningskurve av denne ODE'en ved å utnytte at **Maple klarer å løse mange ODE' er numerisk** (dersom man også oppgir passende initialbetingelser) ved følgende variant av dsolve-kommandoen:

```
[> dsolve( {"ODE for y(x)", "initialbetingelse(r) for y(x)"}, y(x), numeric);
```

Svaret på denne kommandoen er en såkalt "prosedyre", og man bør gi et navn til denne, la oss si "solnum". Vi kan da få skissert grafen til løsningen over intervallet [a,b] ved å gi kommandoen

```
[> odeplot ( solnum, [x , y(x)], a .. b);
```

(som er tilgjengelig hvis plots-pakken er lastet ned, noe vi har gjort tidligere).

Observer at det er ingen "x=" foran a ' en i denne kommandoen!

Vi illustrer dette :

```
[ > solnum:=dsolve({D(y)(x)=sin(y(x)) + x, y(0) =2},
  y(x),numeric);
                                     solnum := proc(rkf45_x) ... end

[ > bilde4:= odeplot(solnum, [x,y(x)], -1.5..1.5 ):
  # Vi gir et navn til grafen til løsningen, for å kunne
  # få det frem sammen med retningsvektorfeltet!

  display({bilde3,bilde4}):
```

For å beregne numerisk koordinatene til f.eks punktet (0.5, y(0.5)) på løsningskurven kan vi gjøre det enkelt slik:

```
[ > solnum(0.5);  
                                     [x = .5, y(x) = 2.510494527302638]
```

Oppgave 6:

Betrakt ODE'en $y' = e^{-y} + x^2$.

- Tegn retningsvektorfeltet til denne likningen i rektanglet bestemt av $x = -1..1$, $y = -1..1$.
- Skisser grafen til løsningen av denne ODE'en med initialbetingelsen $y(0)=0$ for $x = -1..1$ i en tegning som også viser retningsvektorfeltet.

DEL 4

Maple kan også brukes til å gjøre beregninger i lineær algebra. Til det trenger vi å laste inn tilleggsapakken

```
[ > with(linalg):  
Warning, new definition for norm  
Warning, new definition for trace
```

1. Vektorer i Maple skrives på samme måte i alle dimensjoner. Her har vi definert to 5-dimensjonale vektorer

```
[ > v1:=vector([2,-1,0,1,0]); v2:=vector([1,1,0,0,2]);  
                                     v1 := [2, -1, 0, 1, 0]  
                                     v2 := [1, 1, 0, 0, 2]
```

og Maple kan enkelt regne ut prikkproduktet av dem, ved hjelp av den instruktive betegnelsen [**> dotprod**

```
[ > dotprod(v1,v2);  
                                     1
```

Dersom vi prøver å ta prikkproduktet av to vektorer av forskjellig lengde, vil Maple gi oss tilbake en feilmelding om at vektorene ikke er kompatible.

2. Vektorregningen er bare en liten del av lineær algebra. Maple kan også regne med matriser. Det er to måter å definere en matrise. Den ene er ved kommandoen [**> matrix(3,3,[1,0,0, 0,1,0,0,0,1])**]; som gir oss en 3x3-enhetsmatrise. De tor første tallene gir størrelsen og i klammene skriver vi de ni tallene. Den andre måten å angi matrisen på er ved kommandoen [**> array ([1,0,0], [0,1,0],[0,0,1])**]; I dette tilfellet trenger vi ikke angi størrelsen, Maple skjønner det ut i fra de angitt vektorene. Følgende eksempel illustrerer disse to metodene:

```
[ > matrix(2,3,[1,2,-3,0,2,-1]); array([[1,2,-3],[0,2,-1]]);
```

$$\begin{bmatrix} 1 & 2 & -3 \\ 0 & 2 & -1 \end{bmatrix}$$

Som vi ser i den første skriveformen, så angir det første tallet antall rader, mens det andre gir antall kolonner. Dersom vi skulle ha behov for å angi en matrise uten å fastsette verdien på entriene, så kan vi legge inn en blank matrise ved å bruke den første formen, dog uten å angi noen verdier på entriene, altså ved

```
> matrix(3,3);
```

$$\begin{bmatrix} ?_{1,1} & ?_{1,2} & ?_{1,3} \\ ?_{2,1} & ?_{2,2} & ?_{2,3} \\ ?_{3,1} & ?_{3,2} & ?_{3,3} \end{bmatrix}$$

Multiplikasjon av matriser utføres med kommandoen [**>multiply**], hvor input er to eller flere matriser av rett størrelse i forhold til hverandre.

```
> A:=matrix(2,3,[1,2,2,1,0,0]); B:=matrix(3,1,[2,1,0]);
```

$$A := \begin{bmatrix} 1 & 2 & 2 \\ 1 & 0 & 0 \end{bmatrix}$$

$$B := \begin{bmatrix} 2 \\ 1 \\ 0 \end{bmatrix}$$

```
> multiply(A,B);
```

$$\begin{bmatrix} 4 \\ 2 \end{bmatrix}$$

Den samme kommandoen kan brukes for å multiplisere en matrise og en vektor, [**>multiply(M,v)**]; Maple returnerer i det tilfellet en vektor.

For kvadratiske matriser kan vi også regne ut den inverse matrisen. Dette gjør vi ved hjelp av kommandoen [**>inverse(M)**]; Det er selvfølgelig påkrevet at matrisen har en invers, dvs. at determinanten er forskjellig fra 0. Hvorvidt det er tilfelle kan vi sjekke ved kommandoen [**>det(M)**]; Vi illustrerer dette med et eksempel,

```
> M:=matrix(2,2,[1,2,1,-1]);
```

$$M := \begin{bmatrix} 1 & 2 \\ 1 & -1 \end{bmatrix}$$

```
> det(M);
```

$$-3$$

```
> N:=inverse(M);
```

$$N := \begin{bmatrix} \frac{1}{3} & \frac{2}{3} \\ \frac{1}{3} & \frac{-1}{3} \end{bmatrix}$$

og her vil selvfølgelig produktet [**>multiply(M,N)**] gi oss identitetsmatrisen

```
> multiply(M,N);
```

$$\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

Nå kan vi også beregne egenverdier og egenvektorer for en matrise, ganske enkelt ved å bruke kommandoene [**>eigenvalues(M)**]; og [**>eigenvectors**]; Disse vil for en nxn-matrise returnere en liste over n egenverdier, henholdsvis egenvektorer.

```
> eigenvalues(M);
```

$$\sqrt{3}, -\sqrt{3}$$

```
> eigenvectors(M);
```

$$[\sqrt{3}, 1, \{[\sqrt{3} + 1, 1]\}], [-\sqrt{3}, 1, \{[-\sqrt{3} + 1, 1]\}]$$

[>

Denne siste kommandoen returnerer flere størrelser, det første er egenverdien, den andre er multiplisiteten til denne egenverdien og vektoren er den aktuelle egenvektoren.